


Лабораторна робота №1

Моделювання електронних компонент та систем за допомогою SPICE.

Теоретичні відомості

Математичний апарат та основні алгоритми аналізу електронних схем



Електронним компонентом називають частину простору, у якій присутні електромагнітні поля, матеріали та носії заряду та відбуваються електронні процеси. Властивістю електронного компонента є те, що на поверхні, що його обмежує, виділяють скінченну кількість перетинів скінченної площі і припускають, що потік потужності через решту обмежуючої поверхні дорівнює нулю. Ці перетини називають портами або брамами компонента.

До портів як правило вважають під'єднаними лінії передачі, які здійснюють перенос потужності між портами різних компонентів. Відносно підведених ліній передачі порти мають сенс площин відрахунку фаз.

Перенос потужності у лініях передачі може моделюватись на різних рівнях абстрагування:

1. на фізичному рівні, виразами для комплексних амплітуд падаючої та відбитої електромагнітних хвиль;
2. на рівні переносу потужності, виразами для комплексних амплітуд падаючої та відбитої хвиль напруги;
3. на рівні повних нормованих характеристик, виразами для повного нормованого струму та напруги.

Якщо лінія передачі i , відповідно порт, відносяться до типу "Т", то вони мають два незалежні провідники, які називають полюсами, і повні нормовані характеристики переносу у лінії передачі пов'язані із фізичними струмом та напругою між полюсами порта. Якщо лінія передачі не відноситься до типу "Т", то у ній поширюються кілька мод коливань, що переносять потужність. Кожна така мода може бути модельована віртуальною лінією передачі типу "Т" і відповідним віртуальним портом. Компонент, у якого всі порти виражені через полюси, називають багатополюсником, або багатобрамником. Наприклад, компонент із двома портами називається чотрьохполюсником, або двобрамником.

Для систем із зосередженими сталими вважають, що ефективні розміри компонента та довжини ліній передачі значно менші за довжину хвилі і фази сигналів на усіх портах з'єднаних компонентів вважають однаковими. Таку систему моделюють, як декілька зосереджених компонент, полюси яких поєднані ідеальними провідниками. Така модель електронної схеми є предметом розгляду схемотехніки.

Для електронної схеми формулюється та розв'язується основна задача схемотехнічного аналізу - знайти усі струми та усі напруги у схемі. При цьому напругами називають різниці потенціалів між полюсами портів багатополісників, а струмами - ті струми, що втікають або витікають з портів багатополісників.

Застосувавши закони збереження до такої моделі отримують системи рівнянь, що пов'язують між собою групи струмів, що збираються у точках перетину полюсів, або напруг, що обігають контури між полюсами. Відповідно, виділяють найбільш поширені методи аналізу електронних схем - метод контурних струмів та вузлових потенціалів. Рівняння, що формують ці два методи називаються рівняннями Кірхгофа.

Залежності між струмами та напругами на полюсах багатополісного компонента визначаються електронними процесами, що протікають у нього всередині и не є предметом розгляду на даному рівні абстрагування. Ці залежності формулюють у вигляді системи компонентних рівнянь. Фактично ця система і є моделлю електронного компонента. Вхідними та вихідними векторами для неї будуть вектори струмів та напруг на полюсах портів компонента (компонентні струми та напруги), а коефіцієнти при рівняннях залежать від конкретних кількісних фізико-топологічних характеристик даного компонента. Множина цих параметрів є тим, що відрізняє між собою моделі компонентів одного типу і призначення цих параметрів називають ідентифікацією моделі компонента.

Таким чином, задача схемотехнічного аналізу зводиться до розв'язання системи рівнянь Кірхгофа для схеми сумісно із системами компонентних рівнянь моделей компонентів, що вона їх містить.

Найбільш відомим та поширеним засобом для розв'язання такої задачі є система SPICE.

SPICE (англ. Simulation Program with Integrated Circuit Emphasis) — програма симуляції аналогових схем і цифрової логіки, описаної мовою SPICE. Перша версія програми була розроблена в університеті Каліфорнії у 1978 році і призначалась для моделювання електронних схем на великих ЕОМ під ОС UNIX. Згодом програма була перероблена для персональних комп'ютерів фірмою MicroSim, яка згодом була придбана Cadence Design Systems. Архітектура програми та мова опису електронних схем виявилась настільки вдалою, що досі є стандартом де-факто у моделюванні аналогово-цифрових схем. У нинішній час відомі кілька десятків поширених програм автоматизованого проектування електронних

схем, що наслідують концепції SPICE. У даній роботі використовується реалізація PSpice під ОС Windows, або сумісна з нею реалізація GNU Spice під ОС Linux.

Архітектура та налаштування SPICE

SPICE складається з трьох модулів, що діють послідовно:

- парсер - аналізує синтаксис скрипту, що поланий на вхід системи, перевіряє його на помилки та готує до розв'язання систему рівнянь відповідно до завдання;
- солвер - розв'язує систему рівнянь наближено, застосовуючи чисельні методи відповідно до типу системи;
- пробер - виконує обробку результатів розв'язання системи та їх презентацію.

Для налаштування системи використовують скрипт SPICE, який містить опис топології модельованої схеми та задання до моделювання. Також SPICE містить бібліотеки абстрактних моделей електронних компонент та засоби їх ідентифікації.

Синтаксис скрипту SPICE є текстовим файлом, як правило із розширенням ".cir" та має такі загальні особливості:

- перший рядок файлу відведений для назви завдання;
- кожна директива розміщується у окремому рядку, за необхідності переносу використовують символ "+" у першій позиції перенесеного рядка;
- директива починається з першої позиції рядка;
- великі та малі літери не розрізняються;
- імена компонентів та вузлів можуть містити літери, цифри та підкреслення, починаються з літери;
- знак "*" у першій позиції рядка визначає рядок, як коментар

Усі директиви скрипта поділяють на топологічні, директиви налаштування та директиви моделювання. Перші починаються з літери, останні дві категорії - з точки.

Опис топології у SPICE

Топологія схеми у SPICE описується відповідно до вузлової моделі. Вузлам схеми необхідно дати імена, кілька імен вузлів зарезервовано, зокрема "0" - загальний вузол.

Топологічні директиви мають такий вигляд:

(ім'я компонента) (перелік вузлів) (модель)

Модель компонента може бути визначена двома способами - тривіальною та нетривіальною.

Тривіальна модель - це модель, яка має тільки один параметр, тоді замість моделі одразу вказують значення цього параметра. Наприклад опір - тривіальна модель резистора, яка має один параметр - власне омічний опір, записується таким способом:

```
R1 1 2 100
```

- це означає, що резистор "R1" включений між вузлами "1" та "2" і має опір 100 Ом.

Нетривіальна модель визначається через ідентифікацію параметрів моделі класу компонентів. У топологічній директиві вказують ім'я конкретної моделі, що була ідентифікована окремо.

Перелік та послідовність вузлів, до яких під'єднуються полюси компонента, залежить від класу.

Деякі компоненти можуть мати і тривіальні і нетривіальні моделі - наприклад тривіальний опір та модель інтегрального резистора, тривіальна ємність та модель інтегрального конденсатора тощо.

Ім'я компонента має мнемоніку за першою літерою відповідно до класу моделі.

Перша літера	Клас моделі	Опис	Послідовність вузлів
R	r	опір	
L		індуктивність	
C	c	ємність	
D	d	діод	анод, катод
Q	npn pnp	біполярний транзистор	колектор, база, емітер
J	njfet pjfet	польовий транзистор, керований переходом	сток, заслін, витік
M	nmos pmos	МДП транзистор	сток, заслін, витік, підкладка
B	nmf pmf	Арсенід-галієвий польовий транзистор із заслоном Шоткі	сток, заслін, витік
Z	igbt	Статично індукований біполярний транзистор	колектор, заслін, емітер
S W	sw	Ключі, керовані струмом чи напругою	
T O		Лінії передачі без втрат та із втратами	
U		Цифрова схема	
X		Макромодель	

Числові константи у SPICE можна записувати за допомогою масштабних суфіксів:

T	-тера
G	-гіга
meg	-мега
k	-кіло
m	-мілі
u	-мікро
n	-нано
p	-піко
f	-фемто

Опис моделей багатополюсників у SPICE

Нетривіальні моделі у SPICE подають директивами ідентифікації їх параметрів:

```
.model <ім'я моделі> <клас моделі> ( <ім'я параметра>
=<значення> <ім'я параметра>=<значення> ... )
```

Ім'я моделі використовується у топологічних директивах. Перелік параметрів моделі залежить від її класу. Наприклад модель біполярного транзистора КТ315А має наступний вигляд:

```
.model kt315a      NPN      (Is=2.82f  Xti=3  Eg=1.11  Vaf=3.32K  Bf=55
Ne=2.35
+   Ise=554.1p  Ikf=18.04m  Xtb=1.5   Br=.6455   Nc=2   Isc=0  Ikr=0
Rc=0
+   Cjc=17p    Vjc=.75    Mjc=.333   Fc=.5    Cje=8.55p   Vje=.75
Mje=.333
+   Tr=3.36u  Tf=1.02n  Itf=0  Vtf=0  Xtf=0)
```

Більшість виробників електронних компонент, такі, як Motorola, Philipps, National Semiconductor, Harris публікують SPICE-моделі компонент разом із документацією.

Блоки опису моделей компонент можна зібрати в окремому файлі та представити у вигляді бібліотеки компонент. Для підключення бібліотеки використовують директиву .LIB.

Макромоделі SPICE

Макромоделлю у SPICE називають ділянку схеми, побудовану із стандартних компонент, що має внутрішню топологію та йменування вузлів і може бути представлена у вигляді багатополюсного компонента та застосована у інших ділянках схеми.

Опис моделі починається директивою .SUBCKT за якую слідує перелік вузлів у внутрішньому йменуванні, ікі призначені для підключення макромоделі до інших ділянок схеми. Завершується опис директивою .ENDS.

При включенні макромоделі до топології схеми використовують мнемонічну літеру X. За ім'ям компоненту слідує перелік вузлів у йменуванні основної схеми у тій кількості та послідовності, що вони відповідають вузлам макромоделі, переліченим у її описі. За вузлами вказують ім'я макромоделі.

Наприклад макромодель операційного підсилювача К140УД7 має наступний вигляд:

```
* k140ud7 operational amplifier "macromodel" subcircuit
* created using Parts version 1.04 on 10/08/90 at 09:23
*
* connections:   non-inverting input
*                | inverting input
*                | | positive power supply
*                | | | negative power supply
*                | | | | output
```

```

*           | | | | |
.subckt k140ud7 1 2 3 4 5
*
c1  11 12 2.887E-12
c2   6  7 20.00E-12
dc   5 53 dx
de  54  5 dx
dlp 90 91 dx
dln 92 90 dx
dp   4  3 dx
egnd 99  0 poly(2) (3,0) (4,0) 0 .5 .5
fb   7 99 poly(5) vb vc ve vlp vln 0 42.44E6 -40E6 40E6 40E6 -
40E6
ga   6  0 11 12 100.5E-6
gcm  0  6 10 99 31.79E-9
iee 10  4 dc 14.14E-6
hlim 90  0 vlim 1K
q1  11  2 13 qx
q2  12  1 14 qx
r2   6  9 100.0E3
rc1  3 11 9.946E3
rc2  3 12 9.946E3
re1 13 10 6.189E3
re2 14 10 6.189E3
ree 10 99 14.14E6
ro1  8  5 50
ro2  7 99 25
rp   3  4 32.36E3
vb   9  0 dc 0
vc   3 53 dc 3
ve  54  4 dc 3
vlim  7  8 dc 0
vlp 91  0 dc 20
vln  0 92 dc 20
.model dx D(Is=800.0E-18)
.model qx NPN(Is=800.0E-18 Bf=100)
.ends

```

Основні типи аналізу та відповідна організація джерел у SPICE

Джерела напруги та струму у SPICE бувають незалежними (генераторами) та лінійно залежними. Іменування джерел також є мнемонічним за першою літерою:

V	-незалежне джерело напруги
I	-незалежне джерело струму
E	-джерело напруги кероване напругою
F	-джерело струму кероване струмом
G	-джерело струму кероване напругою
H	-джерело напруги кероване струмом

SPICE підтримує кілька типів аналізу електронних схем, основними з яких є статичний аналіз (у сталому струмі), аналіз у частотній області та аналіз у часі. Для кожного типу аналізу певним чином задаються джерела та директиви моделювання.

Аналіз для сталого струму.

Цей режим призначений для моделювання статичних характеристик, коли напруги чи струми певних джерел змінюються стаціонарно і задача схемотехнічного аналізу розв'язується для кожного стану джерел окремо.

Для моделювання у такому режимі використовують директиву:

```
.DC <джерело> <варіація> <джерело> <варіація> ...
```

Тут для кількох джерел можуть бути вказані варіації значення, причому моделювання відбудеться для першого значення останнього джерела і всіх значень попередніх, наступного значення останнього джерела і всіх значень попередніх і т.д.

Варіації можуть бути вказані у формі

```
<мінімум> <максимум> <крок>, або LIST <знач1> <знач2>  
...
```

Наприклад, необхідно побудувати сімейство вихідних характеристик біполярного транзистора КТ315А. Нехай база транзистора підключена до вузла "b", колектор - до вузла "c", а емітер - до загального. До колектора також під'єднаємо джерело напруги Vd, а до бази - джерело струму Ib. Тоді скрипт матиме вигляд:

```
.model kt315a NPN (Is=2.82f Xti=3 Eg=1.11 Vaf=3.32K Bf=55  
Ne=2.35  
+ Ise=554.1p Ikf=18.04m Xtb=1.5 Br=.6455 Nc=2 Isc=0 Ikr=0  
Rc=0  
+ Cjc=17p Vjc=.75 Mjc=.333 Fc=.5 Cje=8.55p Vje=.75  
Mje=.333  
+ Tr=3.36u Tf=1.02n Itf=0 Vtf=0 Xtf=0)  
Q1 c b 0 kt315a  
Vc c 0 1  
Ib b 0 .1  
.DC Vc 1m 15 .1 Ib LIST .1m .2m .5m  
.PROBE  
.END
```

Аналіз у частотній області.

Цей режим призначений для моделювання частотних характеристик схеми. Для натурального експерименту з вимірювання частотних характеристик, зазвичай використовують свіп-генератор і осцилографічний пристрій, що пов'язаний з ним розгорткою. SPICE моделює частотні характеристики у аналогічній спосіб.

Одне і тільки одне джерело напруги у схемі може бути визначено, як свіп-генератор. Для цього використовують суффікс AC, після якого вказують амплітуду генератора та початкову фазу коливань, наприклад

```
V1 1 2 AC 10m 0
```

Крім генератора, у схемі може бути використана будь-яка кількість статичних джерел.

Варіацію частоти генератора задають директивою

.AC LIN <точок> <мінімум> <максимум>

.AC DEC <точок> <мінімум> <максимум>

.AC OCT <точок> <мінімум> <максимум>

.AC POI <знач1> <знач2> <знач3> ...

У інтервальних варіаціях першим параметром задають кількість точок обрахунку на весь діапазон на декаду чи на октаву логарифмічної шкали відповідно. Додатково, на кожному кроці частоти можна задати варіацію будь-якого параметра у схемі за допомогою суфікса SWEEP директиви .AC.

Результатом аналізу буде розв'язання системи рівнянь моделі у комплексній області. Всі струми та напруги будуть подані на подальшу обробку у вигляді функцій частоти у заданому діапазоні.

Аналіз у шумових характеристик.

Аналіз шуму як правило проводять сумісно із аналізом у частотній області і вважають показником шуму спектральну щільність вихідної шумової напруги.

Для аналізу використовують директиву

.NOISE V(<вузол>) Vx

тут вузол (або пара вузлів) вказує на вузлову напругу, яку вважаємо вихідною, а Vx – джерело вхідної шумової напруги.

Для презентації можна використовувати такі вихідні змінні:

NFI [DB] (<компонент>) - флікер-шум компонента (шум 1/f), спричинений стохастичними змінами характеристик матеріалу та поверхневих станів

NSI [DBC] (<компонент>) – дробовий шум спричинений взаємодією електронів з омичними та іншими переходами

NR [BCDEGS] (<компонент>) – тепловий шум спричинений тепловим рухом носіїв заряду

NR [LO | HI] (<компонент>) – тепловий шум опорів входів цифрових пристроїв

NTOT (<компонент>) – сумарний шум компонента

NTOT (ONoise) – сумарний шум схеми

V (ONoise) – усереднений вихідний шум схеми

V (INoise) – еквівалентний вхідний шум схеми

Аналіз у часі.

Цей режим призначений для моделювання динамічних характеристик електронних схем. Для аналізу застосовуються джерела, сигнали яких задаються функціями часу. Функції визначаються через набори параметрів:

```
Vxx <вузол1> <вузол2> <функція> (<парам1> <парам2> ...  
)
```

Функції передбачені такі:

```
pulse( V1 V2 td tr tf Tt T )
```

- *прямокутний імпульс*

V1 - напруга проміжку

V2 - напруга імпульсу

td - час затримки

tr - передній фронт

tf - задній фронт

Tt - тривалість вершини

T - період

```
sin ( Vo Va f td theta phi )
```

- *синусоїдальний сигнал*

Vo - стала складова

Va - амплітуда

f - частота

td - час затримки

theta - коефіцієнт згасання

phi - початкова фаза

```
exp ( V1 V2 td1 tau1 td2 tau2 )
```

- *експоненціальний сигнал (заряд-розряд)*

V1 - нижня межа напруги

V2 - верхня межа напруги

td1 - затримка переднього фронту

tau1 - стала часу переднього фронту

td2 - затримка заднього фронту

tau2 - стала часу заднього фронту

```
pwl ( t1 v1 t2 v2 .... ) <R=repeat> <TD=delay>
```

- *полігональний сигнал*

t1,t2,v1,v2... - точки час-значення

repeat - опціонально період повторення сигналу

delay - опціонально час затримки

```
sffm ( Vo Va Fc MDI Fs )
```

- *частотно-модульований сигнал*

Vo - стала складова

Va	- амплітуда несучої
Fc	- частота несучої
MDI	- глибина модуляції
Fs	- частота модулюючого сигналу

am (Sa Fc Fm Oc Td)

- амплітудно-модульований сигнал

Sa	- амплітуда сигналу
Fc	- частота несучої
Fm	- частота модулюючого сигналу
Oc	- стала складова
Td	- час затримки

Аналіз у часовому просторі передбачає розв'язання системи рівнянь моделі у вигляді диференціальних рівнянь відносно часу. Директива моделювання задає проміжок часу для розрахунку та параметри, що впливають на чисельні методи розв'язання рівнянь.

.TRAN/OP <tstep> <tstop> <tstart> <tmax> <UIC>

tstep	- крок виводу результатів інтегрування ркгулярною сіткою
tstop	- ширина вікна інтегрування
tstart	- опціонально ненульовий час початку інтегрування
tmax	- опціонально обмеження збільшення кроку інтегрування
UIC	- ознака того, що в якості початкових наближень беруться значення, вказані у топологічних директивах суфіксами IC.

Опціональний суфікс /OP означає, що перед початком моделювання буде проведений аналіз статичних характеристик і отримані значення будуть використані у якості початкових наближень.

Завдання

1. Обрати модель операційного підсилювача відповідно до номеру варіанту та знайти її у бібліотеці.
2. Обрати схему включення підсилювача, бажаний коефіцієнт підсилення та напругу живлення відповідно до номеру варіанту.
3. Розрахувати параметри зворотнього зв'язку підсилювача та скласти скрипт SPICE, що описує його топологію.
4. Виконати моделювання у сталому струмі та побудувати передаточну характеристику підсилювача.
5. Виконати моделювання у частотній області та побудувати амплітудно-частотну та фазо-частотну характеристики підсилювача.

6. Виконати моделювання у часовій області та розрахувати перехідну характеристику підсилювача при проходженні прямокутного імпульса. Побудувати епюри вихідної напруги та оцінити час встановлення вихідного сигналу.

Варіанти

№ варіанту	Операційний підсилювач	Тип включення	Коефіцієнт підсилення	Напруга живлення
1	MC33072	неінвертуючий	60	±12
2	LF441	неінвертуючий	990	±9
3	MC33272	інвертуючий	370	±15
4	LF444	інвертуючий	450	±15
5	LM358	неінвертуючий	940	±12
6	MC33178	інвертуючий	170	±6
7	MC33174	інвертуючий	210	±12
8	MC33184	неінвертуючий	80	±9
9	MC33071	неінвертуючий	420	±12
10	MC33284	неінвертуючий	130	±12
11	LF412	інвертуючий	560	±15
12	LM324	неінвертуючий	380	±9
13	MC34081	неінвертуючий	210	±12
14	MC33078	неінвертуючий	510	±6
15	MC33274	інвертуючий	470	±9
16	MC33172	інвертуючий	650	±6
17	MC33282	неінвертуючий	520	±15
18	MC33074	інвертуючий	420	±9
19	LF411	інвертуючий	80	±6
20	LM2902	неінвертуючий	710	±6
21	LM833	інвертуючий	50	±6
22	MC33181	інвертуючий	150	±15
23	MC33171	неінвертуючий	400	±9

24	LM2904	інвертуючий	790	±15
25	LF442	інвертуючий	50	±12

Контрольні запитання.

1. Визначення електронного компонента.
2. Рівні абстрагування в моделі компонента.
3. Визначення системи компонентних рівнянь.
4. Рівні визначеності моделі компонента.
5. Три рівні моделі переносу потужності в лінії передачі (коротко).
6. Розподілені та зосереджені електронні кола. Визначення.
7. Розподілені та зосереджені електронні компоненти. Визначення.
8. Властивості багатополюсника: недисипативність.
9. Властивості багатополюсника: пасивність.
10. Властивості багатополюсника: симетрія.
11. Ідеальні і реальні компоненти. Визначення.
12. Елементарні електронні компоненти.
13. Визначення макромоделей.
14. Що таке аналогові обчислення?
15. Ідентифікація параметрів моделей електронних компонентів.
16. SPICE - аналіз для сталого струму. Джерела та рівняння.
17. SPICE - аналіз у частотній області. Джерела та рівняння.
18. SPICE - аналіз у часовій області. Джерела та рівняння.

Література.

1. Хайнеман Р. PSpice. Моделирование работы электронных схем. Пер. с нем. М.: DMK Пресс, 2001. - 336 с.
2. Гупта К., Гардж Р., Чадха Р. Машинное проектирование СВЧ устройств. М.: Р и С, 1987.
3. Моделирование полупроводниковых приборов и технологических процессов. Под ред. Миллера Д. Пер. с англ. Под ред. Гадияка Г.В. М.: Радио и связь, 1989.
4. Сигорский, В.П.; Петренко, А.И. Алгоритмы анализа электронных схем М.: Советское радио; 1976.
5. <http://bwrc.eecs.berkeley.edu/Classes/IcBook/SPICE/>
6. Петраков О. М. Создание аналоговых PSPICE-моделей радиоэлементов. М.:РС. 2004.